Transactions of Illinois Academy of Sciences
(1982), Volume 75, 1 and 2, pp. 137-142

A SIMPLE AND INSTRUCTIVE NON-LINEAR SYSTEM

by
Gary Kessler
Department of Physics
Illinois Wesleyan University
Bloomington, IL 61701

ABSTRACT

A simple intrinsically non-linear oscillator is discussed which has several interesting pedagogical aspects. The system also lends itself to easy solution by both digital and analog techniques.

INTRODUCTION

Most undergraduate texts in mechanics treat the subject of nonlinear oscillations in some way, the most common examples being the physical pendulum or the anharmonic oscillator. In both of these cases the problem reduces to solving an equation of the form

$$\ddot{x} + \omega^2 x = a x^2 + b x^3 + c x^4 \tag{1}$$

which is usually solved by a perturbation approach. In "Classical Dynamics", (Marion, 1970) the author mentions a system which is intrinsically non-linear (see figure 1). Two springs are connected to each other in a straight line with the outer two ends fixed rigidly to some support. At the junction of the two unstretched springs, a mass is attached and pulled aside perpendicular to the longitudinal axis of the springs. It is then let go and allowed to move freely. Marion does show that the equation of motion is given by

$$\ddot{x} = -\left(\frac{k}{mL^2}\right) x^3 \qquad ; x << L \tag{2}$$

which is clearly non-linear. No further discussion of this problem is presented and the text quickly moves on to the discussion of perturbation theory as applied to equation (1). Modest references to this problem are found in the literature (Slee, 1971) and (Thomchick and McKelvey, 1978), but the former uses a virial theorem approach and proceeds to numerical integration without mentioning the elliptic integral solution, while the latter treats to problem much as does Marion.

The problem just described however, does present some very interesting pedagogical items, such as 1) a qualitative analysis through the use of potential energy diagrams, 2) the use of dimensionless variables, 3) an introduction to solution by elliptic integral tables and 4) a comparison of simple digital and analog computer solutions.

SOLUTION OF THE PROBLEM AND DISCUSSION

Since the force varies as $-x^3$ we see that the potential energy $U(x) \sim x^4$ and is a symmetric function. Because the total energy must always be greater than or equal to the potential energy, $E \geqslant U$, the motion must be periodic. Most students will easily recognize this by quickly sketching the function, although to prove it is somewhat more difficult.

Equation (2) may be made more useful by introducing the dimensionless variables $Q = x/A$ and $T = At\left(\frac{k}{mL^2}\right)^{1/2}$ where A is the amplitude thus giving us the simple equation

$$Q'' = -Q^3 \tag{3}$$

which is deceptively simple! A straightforward analysis will lead to elliptic integrals, as we shall see shortly, but it should be pointed out first that most perturbation approaches fail to produce a solution for this is a case where the main term is the perturbation, namely Q^3, and it cannot be treated as if it were small. Students ought to try solutions of the form,

$$Q = A \cos T + B \cos 3T \tag{4}$$

as is usually done in such cases and see for themselves that A and B are really functions of T. Different values can be obtained for limited ranges, but even then the solution depends on the approximations made and the desired accuracy.

In any event, equation (3) may be multiplied by $Q'dT$ and integrated to give

$$\int_1^Q \frac{dQ}{(1 - Q^4)^{\frac{1}{2}}} = 2^{-\frac{1}{2}} \int_0^T dT \tag{5}$$

where the initial conditions $T = O$, $Q = 1$, $Q' = O$ are used to find the constants of integration. The substition $Q = \cos\theta$ then transforms this equation into

$$\int_0^{\cos^{-1}(\frac{x}{A})} \frac{d\theta}{(1 - \frac{1}{2}\,\mathrm{Sin}^2\theta)^{\frac{1}{2}}} = T \tag{6}$$

which is the standard form of an elliptic integral

$$F(\phi,K) = \int_0^{\phi} \frac{d\theta}{(1 - K^2 \sin^2\theta)^{\frac{1}{2}}} \qquad (7)$$

where $K = \sin\psi = 1/\sqrt{2}$, or $\psi = 45°$. For the first quarter cycle that is, $0 \leq Q \leq 1$ this can be evaluated easily using the standard tables. Since undergraduates do this rarely, meeting elliptic integrals only in the physical pendulum and the potential of a disc, the practice on such a simple example is instructive. The results are presented in the first three columns of Table I.

Finally, the basic equation $Q'' = -Q^3$ is equivalent to the two linear first order equations

$$P' = -Q^3$$

$$Q' = P \qquad (8)$$

which are very amenable to solution by the Runge-Kutta method on a digital computer, or by solution on a small analog computer such as the TR-20. These techniques can be applied without much confusion by the student encountering them for the first time. The Runge-Kutta method is so simple to apply that students need not know numerical analysis in order to apply it or to program it. For comparison an ordinary simple harmonic oscillator solution is included in the program as indicated in the appendix. The results of the calculation are shown in Table I while figures 2 and 3 display the analog program and graphical solution respectively.

It is not always apparent to a student that an equation which appears to be simple can indeed be horrendous, or vice versa. The fact that so many nice techniques can be brought to bear with relative ease on a simple equation is of great value in the classroom.

LITERATURE CITED

1. J. B. Marion, "Classical Dynamics" (Academic Press Inc., New York, 1970).
2. F. W. Slee, Am. J. Phys., *39*, 578 (1971).
3. J. Thomchick and J. P. McKelvey, Am. J. Phys., *46*, 40 (1978).

ACKNOWLEDGEMENT

I would like to thank Andrew Rex, and David Botkin for converting the program to Apple II basic.

APPENDIX

Listing of the computer program used to solve equation (3). It is written in BASIC for use on the Apple II computer. The printout also tabulates a simple harmonic oscillator (position and velocity) to allow a useful comparison.

APPENDIX

```
0  HOME
1  DEF  FN R(X) =  INT (100 * X) / 100
2  DEF  FN S(X) =  INT (1000000 * X) / 1000000
9  REM ****************************************************
10  REM ** THIS PROGRAM SOLVES THE COUPLED DIFFERENTIAL
11  REM ** EQUATIONS  P' = -Q↑3  AND  Q' = P USING THE
12  REM ** FOURTH-ORDER RUNGE-KUTTA METHOD.
100  REM ************************************************
1100  REM ** INITIAL CONDITIONS
1200 H = .01:T = 0:Q = 1:P = 0
1300  REM ** H=STEP SIZE                    T=TIME COUNTER
1400  REM ** Q=POSITION                    P=VELOCITY
1500  PRINT "OUTPUT TO PRINTER (Y/N)? ")
1505  GET A$: PRINT A$
1510  IF A$ < > "Y" AND A$ < > "N" THEN  HOME : GOTO 1500
1600  IF A$ = "Y" THEN 6000
2200  HOME : PRINT "STEP SIZE = ";H
2300  PRINT : PRINT
2500  PRINT "        COUPLED EQNS       HARMONIC OSC"
2700  PRINT "TIME POSITION VELOCITY POSITION VELOCITY"
3000  FOR N = 0 TO 200
3100 Y =  COS (T):V =  -  SIN (T)
3200  REM ** Y=POSITION &
3300  REM ** V=VELOCITY FOR
3400  REM ** HARMONIC OSCILLATOR
3500  PRINT  FN R(T); TAB( 6); FN S(Q); TAB( 15); FN S(P);
3510  PRINT  TAB( 24); FN S(Y); TAB( 33); FN S(V)
3600  GOSUB 3700: GOTO 5000
3700 T = T + H
3800 L1 = H * P
3900 K1 = H * ( - 1 * Q ↑ 3)
4000 L2 = H * (P + K1 / 2)
4100 K2 = H * ( - 1 * (Q + L1 / 2) ↑ 3)
4200 L3 = H * (P + K2 / 2)
4300 K3 = H * ( - 1 * (Q + L2 / 2) ↑ 3)
4400 L4 = H * (P + K3)
4500 K4 = H * ( - 1 * (Q + L3) ↑ 3)
4600 D2 = (K1 + 2 * K2 + 2 * K3 + K4) / 6
4700 P = P + D2
4800 D1 = (L1 + 2 * L2 + 2 * L3 + L4) / 6
4900 Q = Q + D1
4950  RETURN
5000  NEXT N
5999  END
6000  REM ** PRINTER OUTPUT
6100  HOME : HTAB 13: PRINT "OUTPUT TO PRINTER"
6200  PRINT  CHR$ (4);"PR#1"
6300  PRINT "STEP SIZE = ";H
6400  PRINT ,"COUPLED EQUATIONS","HARMONIC OSCILLATOR"
6500  PRINT "TIME","POSITION","VELOCITY","POSITION",
6510  PRINT "VELOCITY": PRINT
6600  FOR N = 0 TO 200
6700 Y =  COS (T):V =  -  SIN (T)
6800  PRINT  FN R(T), FN S(Q), FN S(P), FN S(Y), FN S(V)
6900  GOSUB 3700
7000  NEXT N
8000  PRINT  CHR$ (4);"PR#0": END
```

Table I. Column three shows the solution to equation (3) from elliptic integral tables. Column four shows the computer values of Q for the corresponding times T, of column three.

TABLE I

Q = x/A	$\phi\cos^{-1}(x/A)$, (deg.)	T (tables)	Q (computer)
1.0	0.	0.	1.
0.9	25.8	0.46	0.899
0.8	36.9	0.66	0.803
0.7	45.6	0.84	0.699
0.6	53.1	0.99	0.603
0.5	60.0	1.14	0.495
0.4	66.0	1.28	0.398
0.3	72.5	1.42	0.300
0.2	78.5	1.57	0.194
0.1	84.3	1.71	0.095
0.0	90.	1.85	0.004



Figure 1. An intrinsically non-linear system: a) in the equilibrium position and b) displaced a small distance X from equilibrium.



Figure 2. Analog computer program for the solution to equation (3).



Figure 3. Solution to equation (3) recorded directly from an x-y plotter. Note how much more linear this is near the zero's than the corresponding cosine curve.